
program Wasser; 
 uses 
 FPCMacOSAll; 
 
  const 
 TL  = 0.01;     // Afstand tussen Lamellen (m) 
 NL  = 200 ;    // Aantal Lamellen (-) 
 HL  = 2   ;  // Hoogte Lamellen (m) 
 LL  = 0.5 ;  // Lengte Lamellen (m) 
 DL  = 0.005;  // Dikte Lamellen  (m) 
 WF  = 4.1667e-4; // Waterflow       (m3/sec) 
 GF  = 8.3 ;  // Gasflow     (m3/sec) 
 NY  = 200 ;  // Aantal cellen hoog (-) 
 NX  = 100 ;  // Aantal cellen lang (-) 
 GasIN = 9.26e-4;  // Concentratie NH3 instromend gas (mol/m3) 
 pHinit  = 4   ;   // pH instromend water 
 Temp  = 15  ;  // Temperatuur 
 ka  = 6.05e-7;  // zuurconstante NH4plus bij 15 graden (mol/m3) 
 km  = 0.000031;  // mass transfer constante (m/sec) 
 Henry = 1791.7;  // Henry constante volgens hashimoto et albij T=15 graden 
 dt  = 0.001;  // Tijdstapje (sec) 
 n  = 10000;  // Aantal berekeningen (-) 
 
 var 
 NLK  : integer;  // Aantal lamelkanten (-) 
 WFLK : real;   // Waterflow per lamelkant (m3/sec) 
 GFLK  : real;   // Gasflow per lamelkant   (m3/sec) 
  
 H  : real;   // Celhoogte (m) 
 L  : real;   // Cellengte (m) 
 WCelV : real;   // Celvolume waterkant (m3) 
 GCelV : real;   // Celvolume gaskant (m3) 
 A  : real;   // Uitwisselend oppervlak per cel (m2) 
 WFC  : real;   // Waterflow per cel (m3/sec) 
 GFC  : real;   // Gasflow per cel   (m3/sec) 
 VW  : real;   // Verversingsfactor volumeflow/volume cel water (-) 
 VG  : real;   // Verversingsfactor gasflow/volume cel gas (-) 
 WNuit : real;   // Uitflow van molen uit de wasser in de waterfase (mol) 
 GNCuit : real;   // Uitflow van NH3 uit de wasser in de gasfase (mol/m3) 
 step,i,j :integer;  // tellers 
 
 GNH3     : array[0..(NX+1) , 0..(NY+1)] of double; // concentratie NH3 gasfase   (mol/m3) 
 WN    : array[0..(NX+1) , 0..(NY+1)] of double; // molen N in de waterfase    (mol) 
 WNH3 : array[0..(NX+1) , 0..(NY+1)] of double; // concentratie NH3 waterfase (mol/m3) 
 WNH4 : array[0..(NX+1) , 0..(NY+1)] of double; // concentratie NH4 waterfase (mol/m3) 
 WHplus : array[0..(NX+1) , 0..(NY+1)] of double; // concentratie H+ waterfase  (mol/m3) 
 Transfer : array[0..NX , 0..NY] of double;  // transferred ammonia per cel(mol) 
 
begin 
// Per lamelkant basisgrootheden 
 H := HL/NY; 
 L := LL/NX; 
 A := H*L; 
 NLK := 2*NL; 
 WFLK:= WF/NLK; 
 GFLK:= GF/NLK; 
 WFC := WFLK/NX; 
 GFC := GFLK/NY; 
 WCelV:= 0.5*DL*A; 
 GCelV:= 0.5*TL*A; 
 VW  := (WFC*dt)/WCelV; 

 VG  := (GFC*dt)/GCelV; 
 writeln('Celhoogte       = ',H:8:8,' (m)'); 
 writeln('Cellengte       = ',L:8:8,' (m)'); 
 writeln('Uitw.oppervlak  = ',A:8:8,' (m2)'); 
 writeln('WCelvolume      = ',WCelV:8:8,' (m3)'); 
 writeln('GCelvolume      = ',GCelV:8:8,' (m3)'); 
 writeln('Waterflow Cel   = ',WFC:8:8,' (m3/sec)'); 
 writeln('Gasflow Cel     = ',GFC:8:8,' (m3/sec)'); 
 writeln('Verversing water= ',VW:8:8,' (-)'); 
 writeln('Verversing gas  = ',VG:8:8,' (-)'); 
  
  
// Initieren van de gas en waterfase met lege waarden 
  
 for i:= 0 to NX+1 do 
    begin 
       for j:= 0 to NY+1 do 
   begin 
    GNH3[i,j]:=0; 
    WNH3[i,j]:=0; 
    WNH4[i,j]:=0; 
    WHplus[i,j]:=exp(-pHinit*ln(10))*1000;   // let op: x 1000, want mol/m3 
   end; 
    end; 
  
// Initieren van de gas instroom 
 for j:= 1 to NY do  
 begin 
   GNH3[0,j] := GasIN; 
 end; 
   
// Initieren van de waterstroom 
 for i:= 1 to NX do WNH4[i,0] := 0; 
  
 for  i:=1 to NX do          // bereken WN (mol) voor elke cel 
 begin 
  for j:=0 to (NY+1) do WN[i,j]:= ( WNH3[i,j]+WNH4[i,j] )*WCelV; 
 end; 
 WNuit:=0; 
 GNCuit:=0; 
 writeln; 
  
// MAIN calculation loop 
// --------------------------------------------------------------------------------------------------------------- 
 
 for step:= 1 to n do 
 begin 
 // Print tussenresultaten 
 
 writeln('===== Step ',step,' ===== In: gas=',GasIN*GFLK,'mol/sec ====    Uit: water=', 
  WNuit*WFLK/(WCelV*NX),' mol/sec   gas=',GNCuit*GFLK,' mol/sec ==== Rendement=', 
  100-(GNCuit/GasIN*100):4:4,'% ===='); 
 writeln('WNH4 [1,  0]=',WNH4[1,0],' [20,  0]=',WNH4[20,0],' [40,  0]=',WNH4[40,0],' [60,  0]=', 
  WNH4[60,0],' [80,  0]=',WNH4[80,0],' [100,  0]=',WNH4[100,0]); 
 writeln('WNH4 [1,  1]=',WNH4[1,1],' [20,  1]=',WNH4[20,1],' [40,  1]=',WNH4[40,1],' [60,  1]=', 
  WNH4[60,1],' [80,  1]=',WNH4[80,1],' [100,  1]=',WNH4[100,1]); 
 for j:= 1 to 5 do writeln('WNH4 [1,',(j*20):3,']=',WNH4[1,(j*20)],' [20,',(j*20):3,']=',WNH4[20,(j*20)], 
  ' [40,',(j*20):3,']=',WNH4[40,(j*20)],' [60,',(j*20):3,']=',WNH4[60,(j*20)],' [80,',(j*20):3,']=', 
  WNH4[80,(j*20)],' [100,',(j*20):3,']=',WNH4[100,(j*20)]); 
 writeln(''); 
 writeln('GNH3 [0,  0]=',GNH3[0,0],' [1,  0]=',GNH3[1,0],' [20,  0]=',GNH3[20,0],' [40,  0]=',GNH3[40,0], 
  ' [60,  0]=',GNH3[60,0],' [80,  0]=',GNH3[80,0],' [100,  0]=',GNH3[100,0],' [101,  0]=' 
  ,GNH3[101,0]); 
 writeln('GNH3 [0,  1]=',GNH3[0,1],' [1,  1]=',GNH3[1,1],' [20,  1]=',GNH3[20,1],' [40,  1]=',GNH3[40,1], 
  ' [60,  1]=',GNH3[60,1],' [80,  1]=',GNH3[80,1],' [100,  1]=',GNH3[100,1],' [101,  1]=' 
  ,GNH3[101,1]); 
 for j:= 1 to 5 do writeln('GNH3 [0,',(j*20):3,']=',GNH3[0,(j*20)],' [1,',(j*20):3,']=',GNH3[1,(j*20)], 

  ' [20,',(j*20):3,']=',GNH3[20,(j*20)],' [40,',(j*20):3,']=',GNH3[40,(j*20)],' [60,',(j*20):3,']=', 
  GNH3[60,(j*20)],' [80,',(j*20):3,']=',GNH3[80,(j*20)],' [100,',(j*20):3,']=',GNH3[100,(j*20)], 
  ' [101,',(j*20):3,']=',GNH3[101,(j*20)]); 
 writeln(''); 
 writeln('WN [1,  1]=',wn[1,1],' [20,  1]=',wn[20,1],' [40,  1]=',wn[40,1],' [60,  1]=',wn[60,1],' [80,  1]=', 
  wn[80,1],' [100,  1]=',wn[100,1]); 
 for j:= 1 to 5 do writeln('WN [1,',(j*20):3,']=',wn[1,(j*20)],' [20,',(j*20):3,']=',wn[20,(j*20)], 
  ' [40,',(j*20):3,']=',wn[40,(j*20)],' [60,',(j*20):3,']=',wn[60,(j*20)],' [80,',(j*20):3,']=',wn[80,(j*20)], 
  ' [100,',(j*20):3,']=',wn[100,(j*20)]); 
 writeln('WN [1,101]=',wn[1,101],' [20,101]=',wn[20,101],' [40,101]=',wn[40,101],' [60,101]=', 
  wn[60,101],' [80,101]=',wn[80,101],' [100,101]=',wn[100,101]); 
 writeln(''); 
 
 
 // transport water naar beneden met WNH3, WNH4 en WHplus nieuw uitrekenen 
 
 for i:=1 to NX do 
 begin 
  for j:= NY+1 downto 1 do 
  begin 
   WN[i,j]:=WN[i,j] - (VW*WN[i,j]) + (VW*WN[i,(j-1)]);   // stikstof doorstroom 
   WHplus[i,j]:=WHplus[i,j] - (VW*WHplus[i,j]) + (VW*WHplus[i,(j-1)]); // Hplus doorstroom 
   WNH4[i,j]:= WN[i,j]/( WCelV*(1 + (1/  (1+ (WHplus[i,j]/ka) )))); // herverdeling NH4 
   WNH3[i,j]:= (WN[i,j]/WCelV) - WNH4[i,j];     // herverdeling NH3 
  end; 
 end; 
   
// Calculatie wasser uitflow (mol waterfase) 
 WNuit:=0; 
 for i:=1 to nx do WNuit:=WNuit+WN[i,(ny+1)]; 
 
 
// Transport gas in gasrichting en bereken de nieuwe GNH3 
 
 for j:= 1 to NY do 
 begin 
  for i:= nx+1 downto 1 do GNH3[i,j]:=GNH3[i,j]-(VG*GNH3[i,j])+(VG*GNH3[i-1,j]); 
 end; 
 
 
// Calculatie wasser uitflow (mol gasfase) 
 
 GNCuit:=0; 
 for j:=1 to NY do GNCuit:=GNCuit+GNH3[(NX+1),j]; 
 GNCuit:=GNCuit/NY; 
 
 
// Mass transfer uitrekenen en berekenen nieuwe gas en waterfase pH wordt constant genomen 
 
 for i:=1 to nx do 
 begin 
  for j:= 1 to ny do 
  begin 
   transfer[i,j]:=km*A*dt*( (Henry*GNH3[i,j]) - (ka*WNH4[i,j]/WHplus[i,j]) ); // aantal molen transfer 
   GNH3[i,j]:=GNH3[i,j]-(transfer[i,j]/GCelV);      // nieuwe gasconc. 
   WN[i,j]:=WN[i,j]+transfer[i,j];      // nieuwe totale N in waterkant cel 
   WNH4[i,j]:=WN[i,j]/WCelV*(1- (ka/(WHPlus[i,j]+ka)));  // nieuwe verdeling NH3 NH4plus 
   WNH3[i,j]:=(WN[i,j]/WCelV)-WNH4[i,j]; 
  end; 
 end;  
 
end; // Einde main loop van berekeningen 
 
end. // Einde programma 
�


